
第６章

Graf3Dの試用

　当時のQuickDrawに、曲がりなりにも3D描画機能が備わっていたことは、

あまり知られていないかもしれません。私自身も、その存在自体は耳にした

ことがあったものの、実際にプログラミングしてみたことはなく、今回、本書の

執筆を始めるまでの何十年間かは、その名前も含めてすっかり忘却の彼方

のものとなっていました。それが、この章で取り上げる「Graf3D」です。そのネー

ミングからしてLisa由来のものと分かりますが、QuickDraw本体とは別物

で、由来も異なるもののように思えます。Inside Macintoshにも、その存在

も含めて記載されていないこともあり、資料もあまり残っていないのが現状

です。また、QuickDrawのソースコードに含まれていながら、Macintosh

本体のROMには組み込まれておらず、必要に応じてアプリケーションプロ

グラムとリンクして利用するライブラリとして提供されていました。本書で利

用しているTurbo Pascalでも、１つのUNITとして提供されていて、最初か

らTurbo PascalのデフォルトのUNITとして組み込まれています。一通り

Graf3Dの機能と使い方を確認してから、ここでもTurbo Pascalを使って、

実際にGraf3Dを利用したプログラムを動かしてみましょう。

6-1   Graf3Dとは？
　公開されたQuickDrawのソースコードの中に、そのインターフェースに加えて
実装部分のソースコードも含まれていることからすれば、Graf3DはQuickDraw
の一部と考えることも可能です。しかし、すべてPascalで記述されていることや、
主要な変数の型が浮動小数点数を表すREALになっていること、三角関数のSINや
COSを使っていることなど、QuickDraw本体とは、かなり異質な存在となってい
ます。ちなみに、そのPascalのソースコードは500行にも満たないもので、読んで
理解しようと思えば、それほど難しいものでもないでしょう。本書では、やはり原
則に従って、そのソースコードは転載しません。
　ネーミングについても、QuickDrawが当初「LisaGraf」と呼ばれていたことと考
え合わせると、いかにも関連が強そうな「Graf」を含んでいて、起源が古いことを匂
わせています。そのGrafの部分をQuickDrawに置き換えれば、「QuickDraw3D」
になりますが、後に登場した「QuickDraw 3D」とは別物で、技術的にはほとんど、
あるいはまったく関係ないものと考えたほうが無難でしょう。
　Graf3Dについて、Appleの開発者向けのドキュメントに記述が現れるのは、私
が確認している範囲では、初代Macintoshの発売から約１ヶ月後の1984年2月14
日に発行された「Lisa Development System / Internals Documentation」が最
初です。当初のMacintoshには、セルフの開発環境は存在せず、Macintosh用ア
プリケーションはLisa上で開発することになっていたので、「Lisa Development 
System」向けのドキュメントに記述されるのは、何の不思議もありません。ただ
し、すべてPascalで記述されていることを考えても、Lisa用とMacintosh用の
Graf3Dは、共通のものだったと考えられます。そこに記されている文章を、抄訳す
ると、以下のようになります。

３次元グラフィック機能も利用可能です。QuickDraw ルーチンの上で動作する
Graf3D と呼ばれる UNIT によるものです。Graf3D では、本物のパースペクティ
ブの中で、実数の変数とワールド座標系を使って、３次元オブジェクトを描くこ
とが可能となります。

　また、それからほぼ１年後の、1985年2月15日に発行されたMacintosh用アプ
リケーションの開発者向けに提供されたソフトウェアのドキュメント「Software 
Supplement」には、固定小数点数を使ったMacintosh用のGraf3Dについての記
述が登場します。遅くともその時点では、固定小数点ライブラリのFixMathを使う

ように書き換えられたGraf3Dがリリースされていたのでしょう。その記述の一部
を抄訳しておきましょう。

Lisa Workshop でのみ利用可能です。このバージョンは固定小数点演算を使っ
ていて、これまでの Graf3D のバージョンで使われていた浮動小数点演算に比べ
て、より小さく、より速くなっています。

　さらにそのSoftware Supplementの改定版で、1985年7月2日に発行された
「Software Supplement Update」には、Macintosh上のさまざまな環境でGraf3D
の利用が可能となったことが記されています。

Graf3D は、Lisa Workshop や MDS に加え、MDS リンカーを利用する Macintosh
上の他の言語でも利用できるようになりました。

　その改訂版には、Lisa Workshop上のPascalのサンプルプログラムとして、
Boxes、BoxSphere、sineGridという３つが付属していることも明記されています。
実は、これら３つのソースコードは、Lisa用Pascalだけでなく、例えば本書で利用
しているTurbo Pascalにも移植され、Turbo Pascalのプログラムディスクに含ま
れていたのです。移植と言っても、おそらく、プログラム本体にはほとんど変更の
必要はなかったものと思われます。それらのサンプルプログラムがどんなものなの
か、次の節で実際に動かして見てみましょう。
　以上のように、初代Macintoshが発売されてから１年半ほどの間のGraf3Dその
ものの進化や、動作環境の拡がりについて、ざっと把握することができました。上
に挙げたサンプルプログラムもそうですが、実際にGraf3Dが外部の開発者にも利
用されるようになったのは、固定小数点化がなされてからのことでしょう。Turbo 
PascalにUNITとして組み込まれているバージョンも、もちろん固定小数点版の方
です。そういうわけで、QuickDrawのソースコードに含まれている浮動小数点版
からは少し離れてしまうことになりますが、これ以降は、固定小数点版のGraf3D
を中心にして述べていくことにします。

6-2  Graf3Dの純正サンプルプログラムとAPI仕様

　Graf3Dのプログラミングのためのインターフェースを確認する前に、それを
使ってどのようなことが可能なのか、どのような３次元グラフィックを描くことが
できるのか、Apple純正のサンプルプログラムを実際に動かして、感触をつかんで
おくことにしましょう。ここでは、３種の純正サンプルプログラムをTurbo Pascal
に移植したものを動かして、実際の画面を見ていくことにします。

Boxes

　Boxesは、画面に複数
の立体的な箱を複数並
べて描くものです。動
かしてみる前に、ソー
スコードの先頭部分だ
け、初代Macintoshの
画面上で確認しておき
ましょう。

　画面からまず分かることは、このプログラムは、1986年にTurbo Pascal用に移
植されたものであることです。これは、Graf3Dの歴史には関係なく、単にTurbo 
Pascalが1986年に製品として発売されたものであることを反映したものでしょう。
　また、Turbo PascalのプログラムからGraf3Dを利用するためには、Macintosh
アプリケーションとして一般的なUNIT（Memtypes、QuickDraw、OSIntf、
ToolIntf）に加えて、FixMathとGraf3Dという、２つのUNITをUSESで宣言しな
ければならないことも分かります。このうちFixMathは、Toolboxにはじめから定
義されている一般的な32ビットの固定小数点、Fixedや、その演算機能だけでなく、
新たに別の形式の固定小数点数の型と、その演算機能を定義するものです。仮に
ユーザーのプログラム内でFixMathを使っていなくても、Graf3Dを使うためには、
必ずその前に使用を宣言しなければなりません。それは、Graf3D自体がFixMath
に依存しているからです。つまり、Graf3Dを使うには、「FixMath, Graf3D」を、こ

Graf3Dの純正サンプルプログラム、BoxesをTurbo Pascalに移植
したもの先頭部分

の順番のままUSESで宣言しなければならない、ということになります。
　これらの２つのUNIT、FixMathとGraf3Dについては、後で、Turbo Pascal用
のインターフェースを転載しますが、このサンプルプログラムの続きのソースコー
ドを掲載するわけにもいかないので、これを走らせ、動作画面を確認しましょう。
上のプログラムの部分からも分かるように、箱は全部で16個（0..boxCount）表示
されるはずですが、手前の箱
に隠れて、見えない箱もある
でしょう。それらの箱は、グ
リッド状の床の上に並べら
れ、箱の前面は白、上面は薄
いグレー、側面は黒で塗られ
ているように見えます。

　このプログラムは、Graf3Dのサンプルプログラムとして代表的なもので、巧妙
にできています。というのも、Graf3Dで扱える3Dオブジェクトは、実は「点」だけ
で、「線」や「面」はありません。線は、3Dの点（Point3D）の間に直線を引くことで
表現しています。このためには、MoveTo3DやLineTo3Dといったプロシージャが
用意されていますが、それは引数が３軸の座標点になるからであって、実際には内
部でQuickDraw本体のMoveToやLineToを使って描画することになります。ま
たこのプログラムの画面を見ると、３次元空間で面を塗りつぶすことができている
ように思えますが、実際にはMoveTo3DやLineTo3Dを実行する前に、１つの面ご
とにリージョンをオープンしておき、その面のワイヤーフレームの描画が終わった
ら、その２次元のリージョンを閉じて塗りつぶすことで、3Dの面を塗りつぶしてい
るように見せているのです。また、当然ながら隠線処理や隠面処理もないので、面
は、手前から見て正面、上面、右側面の３つしか描いていません。また奥の方の箱か
ら順に描いて、手前の箱で上書きすることで、天然の隠面処理を実現していると考
えることができるでしょう。
　いずれにしてもこのプログラムは、3Dグラフィックライブラリとしては、かなり
機能の限られたGraf3Dの欠点を覆い隠し、限定的なレイアウトながら、効果的な
デモプログラムとなっています。

Graf3Dの純正サンプルプログラム、BoxesをTurbo Pascal
に移植したものの表示画面

BoxSphere

　このプログラムの名前は、
前半が「箱」、後半は「球面」
という意味で、いったいどん
なものが描かれるのか、名前
からは想像しにくいかもし
れません。実際には、箱の中
に、球面を表現したような、
ワイヤーフレームの3D画像
を描くものです。

　箱の方はともかく、球面と言っても、円の直径を少しずつ変化させ、輪切りの縁
のような線で構成された擬似的な球面を描くものです。描く過程もはっきり見える
ので、螺旋状に描いているように勘違いしてしまいそうですが、それぞれ独立した
円を15個ほど、3D空間内でずらしながら重ねるように描きます。１回描き終わる
と、箱と中の球を、少しずつ傾けながら描き続けるので、デモ効果はあるでしょう。
　このプログラムでは、円の輪郭の座標を計算するために、三角関数のSINを使っ
ています。これはFixMathの中に定義されているFracSinというファンクション
を利用しています。これに限らず、FixMathで使えるファンクションについては、
後で簡単に説明します。

SineGrid

　このプログラムは、起伏の
ある床のような、歪んだグ
リッドだけを描くものです。
１回だけ描くと動作が停止
するので、マウスボタンをク
リックして戻ります。

Graf3Dの純正サンプルプログラム、BoxSphereをTurbo 
Pascalに移植したものの表示画面

Graf3Dの純正サンプルプログラム、SineGridをTurbo 
Pascalに移植したものの表示画面

　このプログラムでは、やはりFixMathに定義された FracSinや FracCosと
いった三角関数を利用してグリッドの交点の３次元座標を計算し、それらの間を
MoveTo3DやLineTo3Dでつなぐことで描いています。
　これらのサンプルプログラムが、実際にどのように3Dグラフィックを描いてい
るのかについては、ソースコードを読んでいただくのが最も確実です。勝手にソー
スコードを転載するわけにもいかないので、気になる方は、Turbo Pascalを入手し
て、付属するソースコードを確認し、実際に動かしてみることをお勧めします。

Graf3DのAPI 仕様

　Graf3Dのプログラミングを試す前に、そのAPIを確認しましょう。公開された
QuickDrawのソースコードに含まれている「Graf3D.p」は、「Graf3D」のUNIT
の定義という形式になっていて、その先頭部分には、INTERFACEが置かれ、この
UNITで利用できる型、プロシージャ、ファンクション、定数、変数の名前や形式が
一通り記述されています。そのオリジナルのUNITでは、実数としてPascalの浮動
小数点数REAL型を使っています。しかし、Turbo Pascalに付属するUNITは、そ
れがすべてFixed型に変更されたものとなっています。このFixedは、32ビットの
固定小数点数で、上位16ビットが整数部、下位16ビットが小数部というものでした。
　このオリジナルのREALタイプのGraf3Dと、FixedタイプのGraf3Dのイン
ターフェースを比べると、REALがFixedになった以外、ほとんど違いはないの
ですが、後者には１つだけ追加されたプロシージャがあります。それはInitGrf3D

（「InitGraf3D」ではないことに注意）という、いかにも重要そうなものです。これは、
Boxesなど、Apple純正のサンプルプログラムでも使っているので、Turbo Pascal
が独自に追加したものではなく、Graf3Dを固定小数点化するタイミングあたりで、
Apple自身が追加したものと思われます。実は、必ずしもこの初期化プロシージャ
を使わなくても、Graf3Dを利用したプログラムは動作するのですが、Graf3Dの初
期化に関わるプロシージャの実行と、2DのQuickDrawの初期化に関わるプロシー
ジャの実行の順番、その他の手順によっては、システムエラーが発生するなど、動
作に不安定な部分が見受けられます。そういった初期化の不確実さを緩和するため
に追加されたものなのかもしれません。
　ここではまず、実際に動作環境が用意できている、Turbo Pascalに付属する
Graf3Dのインターフェースを、Turbo Pascalのリファレンス・マニュアルから転
載しておくことにします。

unit Graf3D(-13);

interface

uses MemTypes,QuickDraw,FixMath;

const radConst = 3754936; {radConst 57.29578}

type Point3D=record
        x: fixed;
        y: fixed;
        z: fixed;
    end;

    Point2D=record
        x: fixed;
        y: fixed;
    end;

XfMatrix  = array [0..3,0..3] of fixed;
Port3DPtr = ^Port3D;
Port3D    = record
                GrPort:                     GrafPtr;
                viewRect:                   Rect;
                xLeft,yTop,xRight,yBottom:  fixed;
                pen,penPrime,eye:           Point3D;
                hSize,vSize:                fixed;
                hCenter,vCenter:            fixed;
                xCotan,yCotan:              fixed;
                ident:                      Boolean;
                xForm:                      XfMatrix;
            end;

var thePort3D: Port3DPtr;

procedure InitGrf3D     (globalPtr: Ptr); external;
procedure Open3DPort    (port: Port3DPtr); external;
procedure SetPort3D     (port: Port3DPtr); external;
procedure GetPort3D     (var port: Port3DPtr); external;

procedure MoveTo2D      (x,y: fixed); external;
procedure MoveTo3D      (x,y,z: fixed); external;
procedure LineTo2D      (x,y: fixed); external;
procedure LineTo3D      (x,y,z: fixed); external;
procedure Move2D        (dx,dy: fixed); external;
procedure Move3D        (dx,dy,dz: fixed); external;
procedure Line2D        (dx,dy: fixed); external;
procedure Line3D        (dx,dy,dz: fixed); external;

procedure ViewPort      (r: Rect); external;
procedure LookAt        (left,top,right,bottom: fixed); external;
procedure ViewAngle     (angle: fixed); external;
procedure Identity;     external;
procedure Scale         (xFactor,yFactor,zFactor: fixed); external;
procedure Translate     (dx,dy,dz: fixed); external;
procedure Pitch         (xAngle: fixed); external;
procedure Yaw           (yAngle: fixed); external;
procedure Roll          (zAngle: fixed); external;
procedure Skew          (zAngle: fixed); external;
procedure TransForm     (src: Point3D; var dst: Point3D); external;

function Clip3D         (src1,src2: Point3D; var dst1,dst2: Point): Boolean;
                        external; 
procedure SetPt3D       (var pt3D: Point3D; x,y,z: fixed); external;
procedure SetPt2D       (var pt2D: Point2D; x,y: fixed); external;

　すでに他の環境で3Dグラフィックのプログラミング経験がある人なら、このイ
ンターフェースを見ただけで、Graf3Dのシンプルな機能が把握でき、使い方もだ
いたい想像できるのではないでしょうか。一方、3Dグラフィックに慣れていない人
には、ここに登場するプロシージャやファンクションの概要の説明を読んだだけで
は、自在にプログラミングできるようにはならないかもしれません。以下、大きな
機能別に、ざっと説明していきますが、ここでは3Dグラフィックの基礎について
は省くので、必要に応じて他の文献を当たっていただければと思います。
　APIとして用意されたプロシージャやファンクションの説明に入る前に、１つだ
け補足しておきます。上に示したインターフェースの先頭部分で定義されている定
数、radConstについてです。名前からして、ラジアンについての定数であること
は分かりますが、その値が3754936と言われても、ピンと来ない人も多いでしょう。
しかしヒントが、コメントにあります。それは、57.29578という実数値です。これ
は、180をπで割った値です。ラジアンでは、デグリーの360度がラジアンでは2π
であることから、デグリー値をこの57.29578で割れば、ラジアン値が求まります。
そして、定数値の3754936というのは、実数の57.29578をFixed型の固定小数点
で表現したものを、LONGINTとして見た場合の値なのです。単純に57.29578に
$10000（65536）を掛ければ求まります。
　この定数が用意されているということは、Graf3Dでの角度は、QuickDrawの
ようなデグリーではなく、ラジアンで指定するのだろうと、勘の良い人は考えるで
しょうが、そこに罠が存在します。元の浮動小数点版のGraf3Dでは、確かに角度
はラジアンで指定していましたが、固定小数点版のGraf3Dでは、角度はデグリー
で指定するようになっています。ではなぜ、ここにradConstが定義されているの
かと言えば、それは元のGraf3Dとの互換性を維持するためとしか考えられませ
ん。ただし、後で取り上げるFixMathに含まれている三角関数のファンクション、
FracCOSとFracSINでは、紛らわしいことに角度をラジアンで指定することに
なっています。それらを利用する場合、デグリーをラジアンに変換する必要があれ
ば、このradConstが定数として使えます。その際の割り算には、やはりFixMath
に含まれているFixed同士の割り算、FixDivが使えるでしょう。その結果はFixed
になりますが、FracCOSとFracSINの引数の型はFixedなのです。
　それでは、Graf3Dに定義されているプロシージャ、ファンクションについて、個
別にざっと説明していきます。

初期化とPort3D関連

PROCEDURE InitGraf3D(globalPtr : Ptr);

　本家のQuickDrawのInitGrafに相当する、Graf3D自体の初期化ルーチンです。実装部
分のソースコードのあるREAL版のGraf3Dには用意されていなかったプロシージャなので、
実際に何を実行しているのかは不明ですが、確かなのは、引数として指定したglobalPtrを、現
在のPort3DPtr、つまりthePort3Dとして設定することです。そのため、このプロシージャを
呼び出す際には、必ず@thePort3Dを指定することになっています。このプロシージャは、
QuickDrawをInitGrafによって初期化した直後に呼び出すのがよいでしょう。

PROCEDURE OpenPort3D(port:Port3DPtr);

　InitGraf3Dで指定したthePort3Dが指すPort3Dレコードの中身を、すべてデフォルト値
に初期化します。通常は、InitGraf3Dを呼び出した直後に、このOpenPort3Dを呼び出すこと
になるでしょう。

PROCEDURE SetPort3D(port: Port3DPtr);

　引数のportで指定したPort3DPtrを、現在のPort3Dとして設定します。この後には、さら
にQuickDrawのSetPortを呼んで、そのPort3Dに付随するGrafPtrを、現在のGrafPort
として設定します。これは、複数のPort3Dを使う場合に、描画対象のPort3Dを切り替える際
に使うものと考えられます。

PROCEDURE GetPort3D(var port: Port3DPtr);

　現在のPort3Dへのポインターを取得します。これはPort3Dを切り替える際に、それ以前
の描画対象だったPort3Dを保存しておく際や、Port3Dのレコードのフィールドに直接アク
セスしたい場合に使えます。

ペンの移動と直線の描画

PROCEDURE MoveTo2D(x,y:fixed);

　ペンを、z軸上の座標値は維持したまま、引数で指定した座標値、x、yの点に移動します。

PROCEDURE MoveTo3D(x,y,z:fixed);

　ペンを、引数で指定した座標値、x、y、zの点に移動します。

PROCEDURE Move2D(dx,dy:fixed);

　z軸上の座標値は維持したまま、引数で指定した変位dx、dyを、それぞれ現在の座標値、x、
yに加えて、ペンを移動します。

PROCEDURE Move3D(dx,dy,dz:fixed);

　引数で指定した変位dx、dy、dzを、それぞれ現在の座標値、x、y、zに加えて、ペンを移動し
ます。

PROCEDURE LineTo2D(x,y:fixed);

　現在の点から、z軸上の座標値は維持したまま、引数で指定した座標値、x、yの点まで直線を
描きます。

PROCEDURE LineTo3D(x,y,z:fixed);

　現在の点から、引数で指定した座標値、x、y、zの点まで直線を描きます。

PROCEDURE Line2D(dx,dy:fixed);

　現在の点から、z軸上の座標値は維持したまま、引数で指定した変位dx、dyを、それぞれ現
在の座標値、x、yに加えた点まで直線を描きます。

PROCEDURE Line3D(dx,dy,dz:fixed);

　現在の点から、引数で指定した変位dx、dy、dzを、それぞれ現在の座標値、x、y、zに加えた
点まで直線を描きます。

視点（カメラ）の設定

PROCEDURE ViewPort(r:rect);

　QuickDrawのgrafPortの中の、どの位置に3Dグラフィックを表示するかを、引数の長方
形で指定します。通常はウィンドウ全体（portRect）を指定すればよいでしょう。

PROCEDURE LookAt(left,top,right,bottom:fixed);

　Graf3D空間の中の、どの部分を2D画面に表示するのかを、left、top、right、bottomとい
う４つの座標値で構成する長方形で指定します。それらの型はFixedなので、整数値で座標を
指定したければ、通常はFixRatioを使って、Fixedに変換してから設定します。ここで指定
した範囲が、上のViewPortで指定した2Dの長方形にマッピングされることになります。
このプロシージャを呼んだ後は、すぐにViewAngleを呼んで、視線の角度を指定しなければ
なりません。

PROCEDURE ViewAngle(angle:fixed);

　視点から見るパースペクティブを、視野ピラミッドの水平方向の範囲の角度、angle（デグ
リー）で指定します。指定する値によって、カメラのレンズを交換するような効果が得られる
でしょう。角度として0を指定すれば、パースのまったく付かない望遠レンズのような描画と
なり、25程度が肉眼で見るのに近い値とされています。それより大きい値は、広角レンズのよ
うな効果をもたらします。

座標変換

PROCEDURE Identity;

　座標変換の行列を、単位行列にリセットします。また、Port3Dレコードのidentフィールド
を真にセットします。この状態では、座標変換を施しても、元と同じ座標になります。以下の座
標変換に関するプロシージャの説明は、この単位行列に対して単独に適用した場合の話です
が、実際にはIdentityを実行しない限り、座標変換は行列に累積されていきます。

PROCEDURE Scale(xFactor,yFactor,zFactor:fixed);

　座標変換の行列を、x、y、z成分それぞれに、xFactor、yFactor、zFactorを掛けた値にス
ケーリングするようなものに設定します。

PROCEDURE Translate(dx,dy,dz:Fixed);

　座標変換の行列を、x、y、z成分それぞれに、dx、dy、dzを加えて平行移動するようなものに
設定します。

PROCEDURE Pitch(xAngle:fixed);

　座標変換の行列を、x軸についてxAngleだけ回転させるようなものに設定します。x軸の正
方向から原点を見たとき、正の値の回転は時計回りとなります。

PROCEDURE Yaw(yAngle:fixed);

　座標変換の行列を、y軸についてyAngleだけ回転させるようなものに設定します。y軸の正
方向から原点を見たとき、正の値の回転は時計回りとなります。

PROCEDURE Roll(zAngle:fixed);

　座標変換の行列を、z軸についてzAngleだけ回転させるようなものに設定します。z軸の正
方向から原点を見たとき、正の値の回転は時計回りとなります。

PROCEDURE Skew(zAngle:fixed);

　座標変換の行列を、z軸についてzAngleだけスキュー（せん断変形）させるようなものに設
定します。その座標変換で変更されるのはx座標の値だけで、y、z軸の値は維持されます。z軸
の正方向から原点を見たとき、正の値の回転は時計回りとなります。

PROCEDURE Transform(src:Point3D; VAR dst:Point3D);

　座標変換の行列をsrcで与えられたPoint3Dの点に適用し、結果をdstで示されるPoint3D
の点として返します。

3Dクリッピング

FUNCTION Clip3D(src1,src2:Point3D; VAR dst1,dst2:Point):Boolean;

　引数のsrc1とsrc2で与えられたPoint3Dの２点を結ぶ直線が、視野ピラミッドの範囲に収
まるかどうかを真偽値として返します。もしその直線のどの部分も視野ピラミッドに入らない
場合には偽を、それ以外なら真を返します。また、参照渡しの引数、dst1とdst2には、クリッ
ピング後のsrc1とsrc2の座標を、画面の2D座標に変換したものが入ります。このClip3Dの
動作には、座標変換行列の設定は影響しません。

点の設定

PROCEDURE SetPt2D(VAR pt2D:Point2D; x,y:fixed);

　参照渡しの引数で指定したPoint2Dレコード、pt2Dのフィールドを、それぞれx、yで指定
した値に設定します。

PROCEDURE SetPt3D(VAR pt3D:Point3D;x,y,z:fixed);

　参照渡しの引数で指定したPoint3Dレコード、pt3Dのフィールドを、それぞれx、y、zで指
定した値に設定します。

6-3   FixMathの概要
　ここでは、念のために、Graf3Dも依拠しているFixMathのインターフェースを、
Turbo Pascalのリファレンス・マニュアルから抜粋して掲載しておきます。イン
ターフェースは、すべてファンクションの形式となっています。それらの名前や引
数、戻り値の仕様を見れば、機能や使い方は容易に想像でき、誤解の余地はないで
しょう。ファンクションの定義の後ろに、inline〜と書いてあるものは、それらの
実装部分が初代MacintoshのROMに含まれていて、プログラムの実行時にはト
ラップコードとして実行されるものであることを示しています。そして、external
と書いてあるものは、実装部分は別途ライブラリとして用意されているものを、リ
ンクして使うものと考えればよいでしょう。
　これらのファンクションについて、個別の説明は省きますが、やはりここでも１
点だけ、この中に登場する変数の型について補足しておきます。それは、ここに挙
げる多くのファンクションの引数や戻り値としても指定されている「Fract」です。
その定義も、以下のインターフェースに記述されていますが、それはtype Fract = 
LongInt;というものです。この定義だけでは、形式的には32ビットのLONGINT
として扱うこともできるということで、32ビットの固定小数点型、Fixedと同じで
す。このFractも、やはり固定小数点型の１つなのですが、FixedがLONGINTの上
位16ビットを整数部、下位16ビットを小数部としていたのに対し、Fractは、符号
ビットも含めて上位2ビットが整数部、下位30ビットが小数部という、小数部に特
化した固定小数点数となっています。これは、SINやCOSなどの三角関数と相性が
よいものと想像できるでしょう。LONGINTの値を見ても、それがFractとして実
際にどんな値を表すのか、Fixed以上に分かりにくいのですが、やはりFixMathが
提供しているFix2FracやFrac2Fixといったファンクションによって、Fixedと相
互変換することが可能です。ただし、FractをFixedに変換すると、小数部の精度が
30ビットから16ビットに落ちてしまうことは、頭に入れておく必要があります。　
このFixMathについては、以下にPascalのインターフェース部分を示すに留めま
す。

unit FixMath(-12);

interface

uses MemTypes;

type Fract = LongInt;

{ These routines are only available on a system with a 128K ROM: }

function Long2Fix(x:longint):Fixed;     inline $A83F;
function Fix2Long(x:fixed):LongInt;     inline $A840;
function Fix2Frac(x:fixed):Fract;       inline $A841;
function Frac2Fix(x:fract):Fixed;       inline $A842;

{ Functions to convert between fixed-point types }

function Fix2X(x:fixed):Extended;       inline $A843;
function X2Fix(x:extended):Fixed;       inline $A844;
function Frac2X(x:fract):Extended;      inline $A845;
function X2Frac(x:extended):Fract;      inline $A846;

{ Functions to convert between fixed, fract, and the extended }
{ floating-point type }

function FixAtan2(x,y:LongInt):Fixed;   inline $A818;

{ The following routines are supplied as glue code:

function FracMul(x, y: Fract): Fract;   external;
function FixDiv(x, y: Fixed): Fixed;    external;
function FracDiv(x, y: Fract): Fract;   external;
function FracSqrt(x: Fract): Fract;     external;
function FracCos(x: Fixed): Fract;      external;
function FracSin(x: Fixed): Fract;      external;

6-4	 Graf3Dのサンプルプログラム試作
MacTechサンプルのリソースを復元して動作

　Graf3Dが固定小数点化されて、サードパーティ製の開発環境でも利用できるよ
うになった当時でも、その解説やサンプルプログラムが、それほど潤沢に出回って
いたことはなかったでしょう。それから何十年も経過して、本書の執筆時点（2025
〜2026年初頭）まで残っていて、比較的容易にアクセスできるものとしては、元は
Apple純正のサンプルプログラム３種と、MacTech誌の記事のウェブ上のアーカ
イブくらいのものにとどまっています。MacTech誌は、初代Macintoshの発売か
ら少し遅れて誕生した、主にMacintosh用アプリケーションのプログラミング技
術を扱った専門誌です。今でもその多くの記事をウェブ上で読むことができます。
　「MacTech Graf3D」といったキー ワ ードで 検 索 すると「Understanding 
Graf3D」（Volume 3 / Issue 3）という記事を読むことができるでしょう。これ
が未来永劫存続するか分かりませんが、URLを示しておくと、「http://preserve.
mactech.com/articles/mactech/Vol.03/03.03/UnderstandingGraf3D/index.
html」となっています。
　この記事には、当時のMacintosh用のPascal開発環境として、Turbo Pascalと
人気を二分していたLightspeed Pascal用のプログラムが掲載されています。ただ
し、Turbo Pascal用のUNITのインポート用の記述も、コメントとして記されてい
るので、少ない変更でTurbo Pascalでも動かせるようになっています。とはいえ、
困ったことに、このプログラムで使用するリソースの中身の定義については、どこ
にも記述がありません。おそらく当時は、リンク可能なリソースファイルが、付録
のフロッピーディスクの中に含まれていたのでしょう。また、本来の記事には配置
されていたはずの画像ファイルは、現時点ではリンク切れとなっていて表示されな
いので、どのような画面を表示するプログラムであったかさえ、視覚的に確認する
ことができなくなっています。
　そこで、掲載されているプログラムのソースコードから逆算して、必要なリソー
スを復元して定義し、アプリケーションとして動かしてみることにしました。２つ
のウィンドウそのものや、１つのウィンドウの中のコントロール（スクロールバー
が３つ）の正確な配置は分かりませんが、かなり近いものにはなっていると考えて
います。

まずは、プログラムの動作画面を見てみましょう。

MacTech誌Volume 3/Issue 3に掲載されたGraf3Dのサンプルプログラムを復元した画面

　このプログラムで描画するのは、平面を表すグリッドと、その上に浮かぶ正四面
体のワイヤーフレーム表現だけです。視点の設定によって、プログラム起動時の平
面は、１本の横線にしか見えないでしょう。そのような3Dグラフィックを表示す
るのが「View」ウィンドウですが、その下にもう１つ、「Input」ウィンドウがあり
ます。その中にはX、Y、Zの各軸について、オブジェクト、つまり正四面体の回転量
を指定する「X Rotation」、「Y Rotation」、「Z Rotation」という３本のスクロール
バーが配置されています。スクロールバーのツマミをドラッグして動かすと、それ
に応じてViewウィンドウの中の表示が変化し、正四面体が回転します。これは、起
動直後の、オブジェクトを回転させるモードの動作です。
　ここで「Switch」メニューから「Rotate Space」を選ぶと、今度は空間を回転させ
るモードとなります。このとき、下のInputウィンドウの３本のスクロールバーの
ラベルは「Pitch」、「Yaw」、「Roll」という３つに変わり、スクロールバーを動かすと、
正四面体の下の平面を含めて、空間全体が回転するようになります。
　比較的シンプルなプログラムですが、これによって3Dグラフィックの基礎と、
Graf3Dの機能が把握できる優れたサンプルプログラムだと思います。そこで、同
じようなものを自分で新たに作るより、これを紹介する方が適切だと考え、リソー
スを復元して動かしてみたというわけです。自らの環境で、このプログラムを動
かしてみたいという人は、以下のリソースの定義ファイルを、RMakerを使って
コンパイルして、MacTechのウェブサイトにあるプログラムにリンクしてくださ
い。本書の付録のソースコードの中にも、リソースの定義ファイル「MacTech3D.
R」に加え、それをコンパイルした「MacTech3D.Rsrc」ファイルを含めて、ダウン

ロードできるようにしてあります。それをそのままリンクして使っていただければ、
RMakerによる処理は不要です。このリソースファイルを、Pascalのソースコード
と同じフォルダーに入れ、プログラムの先頭部分に{$R MacTech3D.Rsrc}のよう
な記述を入れておけば、自動的にリンクして、プログラムを動かせるはずです。

MacTech3D.Rsrc

Type MENU
,128
\14
  About Me...
  (-

  ,129
File
  Quit/Q

  ,130
Edit
  Undo/Z
  (-
  Cut/X
  Copy/C
  Paste/V
  Clear

  ,131
Switch
  Rotate Space

Type WIND
  ,32000
View
45 8 245 504
Visible NoGoAway
0
0

  ,32001
Input
271 8 336 504
Visible NoGoAway
0
0

Type CNTL
,128
horizontal scroll bar
25 10 41 155
Visible
16
0
0 50 0

,129
horizontal scroll bar

25 175 41 320
Visible
16
0
0 50 0

,130
horizontal scroll bar
25 340 41 485
Visible
16
0
0 50 0

Type ALRT
  ,128
100 100 190 400
129
CCCC

Type DITL
  ,129
2

StatText
25 20 36 300
This is a Graf3D Test Program.

BtnItem
60 230 80 290
OK

キューブ連続回転プログラムの試作

　最後に、Graf3Dを使ったシンプルなオリジナルのサンプルプログラムを示すこ
とにします。Graf3Dの使い方は、その初期化の手順も含めて、やや分かりにくい部
分がありますし、これまでに見たApple純正から移植したものや、MacTechに掲
載されたものも、凝った部分に覆い隠されて、エッセンスの部分が見えにくくなっ
ているというきらいがあります。そこでここでは、Graf3Dならではの機能を利用
しつつ、とにかくできるだけ短いプログラムにする、ということを心がけました。
また、Turbo Pascalの実行環境のデフォルトのウィンドウをそのまま使うように
したので、初期のMacintoshの標準的なアプリケーションの初期化手順は踏んで
いませんが、それによって、かえってGraf3D特有の初期化部分が見えやすくなっ
ていると考えています。
　プログラムを起動すると、ウィンドウの中で小さいキューブ（正六面体）が回転
し続け、マウスボタンを押すと終了するというだけのものです。まずは、その動作
画面を示します。

できるだけ短いプログラムでGraf3Dならではの機能を実現す
ることを目指したRotatingCubeの動作画面

　この画面からも分かるように、キューブを構成する点の数は８個ですが、それを
描くための直線は、全部で12本必要となります。一般的な手法のように、「辺」を定
義して１本ずつ直線を描くようにすると、始点と終点で合計24の点の3D→2D変
換が必要となり、初代のMacintoshにとっては、それだけでも大きな負荷となりま
す。そこで、このプログラムでは、12本の辺を、できるだけ一筆書きで描くように
してみました。最初に８本を一筆書きで描き、それだけでは描ききれない４本を後
から追加しています。
　それも含めて、あまり美しいプログラムとは言えませんが、動作速度については、
ここから大幅に速くするのは難しいのではないかと考えています。以下にソース
コード全体を示します。複雑なことは何もしていないので、説明は省きます。

  1|  PROGRAM RotatingCube;
  2|
  3|  USES
  4|    MemTypes,QuickDraw,OSIntf,ToolIntf,FixMath,Graf3D;
  5|
  6|  VAR
  7|    verts:    ARRAY[0..7] OF Point3D;
  8|    fTicks:   LONGINT;
  9|    my3DPort: Port3D;
 10|    myPort:   GrafPtr;
 11|    i:        INTEGER;
 12|
 13|  PROCEDURE InitCube;
 14|  BEGIN
 15|    SetPt3D(verts[0],FixRatio(-10,1),FixRatio(10,1),FixRatio(-10,1));
 16|    SetPt3D(verts[1],FixRatio(-10,1),FixRatio(-10,1),FixRatio(-10,1));
 17|    SetPt3D(verts[2],FixRatio(-10,1),FixRatio(-10,1),FixRatio(10,1));
 18|    SetPt3D(verts[3],FixRatio(-10,1),FixRatio(10,1),FixRatio(10,1));

 19|    SetPt3D(verts[4],FixRatio(10,1),FixRatio(10,1),FixRatio(10,1));
 20|    SetPt3D(verts[5],FixRatio(10,1),FixRatio(-10,1),FixRatio(10,1));
 21|    SetPt3D(verts[6],FixRatio(10,1),FixRatio(-10,1),FixRatio(-10,1));
 22|    SetPt3D(verts[7],FixRatio(10,1),FixRatio(10,1),FixRatio(-10,1));
 23|  END;
 24|
 25|  PROCEDURE DrawCube;
 26|  BEGIN
 27|    Moveto3D(verts[0].x,verts[0].y,verts[0].z);
 28|    FOR i := 1 TO 8 DO
 29|      Lineto3D(verts[i MOD 8].x,verts[i MOD 8].y,verts[i MOD 8].z);
 30|    Moveto3D(verts[0].x,verts[0].y,verts[0].z);
 31|    Lineto3D(verts[3].x,verts[3].y,verts[3].z);
 32|    Moveto3D(verts[1].x,verts[1].y,verts[1].z);
 33|    Lineto3D(verts[6].x,verts[6].y,verts[6].z);
 34|    Moveto3D(verts[2].x,verts[2].y,verts[2].z);
 35|    Lineto3D(verts[5].x,verts[5].y,verts[5].z);
 36|    Moveto3D(verts[4].x,verts[4].y,verts[4].z);
 37|    Lineto3D(verts[7].x,verts[7].y,verts[7].z);
 38|  END;
 39|
 40|  BEGIN
 41|    InitGrf3D(@thePort3D);
 42|    Open3DPort(@my3DPort);
 43|    SetPort3D(@my3DPort);
 44|    GetPort(myPort);
 45|    ViewPort(myPort^.portBits.bounds);
 46|
 47|    LookAt(FixRatio(-90,1),FixRatio(50,1),
 48|           FixRatio(90,1),FixRatio(-50,1));
 49|    ViewAngle(FixRatio(40, 1));
 50|
 51|    InitCube;
 52|
 53|    Pitch(FixRatio(1,1));
 54|    Yaw(FixRatio(3,1));
 55|    Roll(FixRatio(5,1));
 56|
 57|    HideCursor;
 58|
 59|    REPEAT
 60|      FOR i := 0 TO 7 DO
 61|        Transform(verts[i],verts[i]);
 62|      EraseRect(myPort^.portRect);
 63|      DrawCube;
 64|      Delay(20,fTicks);
 65|    UNTIL Button;
 66|  END.

